

Използване на формули за обработка на текст в Excel

В този документ ще разгледаме:


- Как Excel обработва текст, въведен в клетките
- Какви функции има за обработка на текст
- Примери за обработка на текст

Работа с текст

Когато се въведат данни в клетка, Excel автоматично определя дали е въведена формула, число, дата, час или всичко останало (текст). В различните пособия вместо текст можем да видим стринг или текстов стринг (синоними са).


Една клетка може да съдържа до 32000 символа (Excel 2019). Не забравяйте обаче, че Excel не е Word и изключително рядко се налага да се ползват огромни текстове в една клетка.

Ако трябва да се покаже голям текст в страница (worksheet) на Excel, може да се ползва Text Box. За целта изберете Insert->Text->Text Box, след което посочете страницата, в която искате да поставите text box и започнете да пишете. Това е най-лесният начин да редактирате голям текст в Excel. Ако трябва да прилагате формули и функции към текста, той трябва да бъде в клетка.



Когато импортирате данни в Excel, може да се сблъскате с често срещан проблем: понякога импортираните данни се определят като текст, а не като числа. Това може да доведе до грешки, при използване в числови формули. Индикаторът за грешка представлява зелен триъгълник в горния ляв ъгъл на клетката като в добавка се показва икона в съседство. Активирайте клетката, изберете иконата и ще получите различни опции за използване на данните. В частност, ако трябва да се преобразуват в числа, изберете Convert to Number от списъка.

За да определите какви проверки за грешки ще действат, изберете File->Options->Formulas tab.



Използване на текстови функции

Excel има голям асортимент от текстови функции. Намират се в Formulas tab->Text. Много от тях не се използват само за текстове, а и за числа. Показаните по-долу примери са от книгата Alexander, Michael; Kusleika, Richard; Walkenbach, John. Excel 2019 Bible. Wiley.

Обединяване на текст (конкатенация)

За обединяване на текст от две или повече клетки се ползва &. Например, обединяване на име и фамилия (не забравяйте празната позиция между двете имена):

| <div> <div> <div></div> <div></div> </div> <div> <div>✖</div> <div>✔</div> </div> <div> <div><i>fx</i></div> <div>=B3&" "&C3 </div> </div> </div> | | |
|---|------------|--------------------|
| B | C | D |
| FirstName | LastName | Full Name |
| Guy | Gilbert | =B3&" "&C3 |
| Kevin | Brown | Kevin Brown |
| Roberto | Tamburello | Roberto Tamburello |
| Rob | Walters | Rob Walters |
| Thierry | Alexander | Thierry Alexander |
| David | Bradley | David Bradley |
| JoLynn | Dobney | JoLynn Dobney |
| Ruth | Ellerbrock | Ruth Ellerbrock |
| Doris | Hartwig | Doris Hartwig |
| John | Campbell | John Campbell |

В Excel 2019 можем да използваме и TEXTJOIN(разделител,игнорирай празните стойности,текст)

Разделител показва какъв символ ще се използва между различните клетки. `игнорирай_празните_стойности` определя какво да се случи, когато в дадена клетка няма нищо: използваме `TRUE`, ако трябва да се игнорира и `FALSE` в противен случай. Текст е адресът на клетките, които трябва да се обединят. Пример – обединяване на име, инициал на презиме и фамилия:

| | A | B | C | D | E |
|----|---|------------------|-----------------------|-----------------|-----------------------------|
| 13 | | | | | |
| 14 | | | | | |
| 15 | | FirstName | Middle Initial | LastName | TEXTJOIN |
| 16 | | Guy | H. | Gilbert | =TEXTJOIN(" ",TRUE,B16:D16) |
| 17 | | Kevin | P. | Brown | Kevin P. Brown |
| 18 | | Roberto | B. | Tamburello | Roberto B. Tamburello |
| 19 | | Rob | A. | Walters | Rob A. Walters |
| 20 | | Thierry | D. | Alexander | Thierry D. Alexander |
| 21 | | David | | Bradley | David Bradley |
| 22 | | JoLynn | | Dobney | JoLynn Dobney |
| 23 | | Ruth | T. | Ellerbrock | Ruth T. Ellerbrock |
| 24 | | Doris | W. | Hartwig | Doris W. Hartwig |
| 25 | | John | | Campbell | John Campbell |

Промяна на начина на писане(главни малки букви)

Excel предоставя три функции за тази цел: UPPER – преобразува само в главни букви, LOWER – преобразува само в малки букви, PROPER – всяка дума започва с главна буква, а останалите букви са малки. За да преобразуваме текста в Sentence case (само първата буква на първата дума е главна, останалите са малки), използваме формула, в която с LEFT(C4,1) взимаме само първата буква на изречението, а с RIGHT(C4,LEN(C4)-1) взимаме останалата част от текста:

| | |
|---|--|
| | The QUICK brown FOX JUMPS over the lazy DOG. |
| =UPPER(C6) | THE QUICK BROWN FOX JUMPS OVER THE LAZY DOG. |
| =LOWER(C7) | the quick brown fox jumps over the lazy dog. |
| =PROPER(C8) | The Quick Brown Fox Jumps Over The Lazy Dog. |
| =UPPER(LEFT(C4,1))&LOWER(RIGHT(C4,LEN(C4)-1)) | The quick brown fox jumps over the lazy dog. |

Функцията LEFT(текст,брой) отделя брой символи, започвайки отляво надясно, функцията RIGHT(текст,брой) отделя брой символи, започвайки отдясно наляво, а LEN(текст) връща дължината на текста в брой символи.

Премахване на празните позиции в текста

Често се налага да се премахнат празните позиции (шпации) в началото и края на текста. За целта използваме функция TRIM(текст), която връща текста без тях. TRIM премахва само празната позиция с код 32, а не с код 160 (т.е. използва се CHAR(32), а не CHAR(160)). Пример:

| Original Text | Trimmed Text |
|---------------|--------------|
| ABCD | ABCD |
| A B C D | A B C D |
| Alan Jones | Alan Jones |
| ABCD | =TRIM(B7) |

Отделяне на част от текста

Една от най-важните функции в обработката на текст е отделянето на части от него. За целта се използват функции LEFT, RIGHT и MID(текст,начало, брой) – отделя част от текста (подстринг), започвайки от позиция начало до начало+брой символи. С тези функции можем да отделяме телефонни номера, пощенски кодове или код на служители. Пример:

| Convert these 9 digit postal codes into 5 digit postal codes. | | |
|---|-----------|---------------|
| Zip | Zip | |
| 70056-2343 | 70056 | =LEFT(A4,5) |
| 75023-5774 | 75023 | =LEFT(A5,5) |
| Extract the phone number without the area code | | |
| Phone | Phone | |
| (214)887-7765 | 887-7765 | =RIGHT(A9,8) |
| (703)654-2180 | 654-2180 | =RIGHT(A10,8) |
| Extract the 4th character of each Job Code | | |
| Job Code | Job Level | |
| 2214001 | 4 | =MID(A14,4,1) |
| 5542075 | 2 | =MID(A15,4,1) |
| 1113543 | 3 | =MID(A16,4,1) |

Намиране на зададен символ в текста

Когато се налага определяне на позицията на търсен символ, се използва функция FIND(шаблон,текст), която търси присъства ли шаблона в текста. Функцията връща позицията на първото срещане на шаблона в текста отляво надясно. Пример:

| Product Code | Extract the Numbers |
|---------------|---------------------------|
| PWR-16-Small | =MID(B3,FIND("-",B3)+1,2) |
| PW-18-Medium | 18 |
| PW-19-Large | 19 |
| CWS-22-Medium | 22 |
| CWTP-44-Large | 44 |

С FIND намираме позицията на първото срещане на "-", след което с MID отделяме двете цифри.

Намиране на второ срещане на зададен символ в текста

За целта се използва третият параметър (той не е задължителен, ако го няма търсенето започва от първия символ в текста), който показва откъде започва търсенето: FIND(шаблон,текст,начало). Ако не знаем коя е точната позиция за начало, се използва вложено извикване на FIND:

| =MID(B3,FIND("-",B3,FIND("-",B3)+1)+1,10000) | | | | | |
|--|---------------|------------------------------|---|---|---|
| A | B | C | D | E | F |
| | Product Code | Extract the Size Designation | | | |
| | PWR-16-Small | Small | | | |
| | PW-18-Medium | Medium | | | |
| | PW-19-Large | Large | | | |
| | CWS-22-Medium | Medium | | | |
| | CWTP-44-Large | Large | | | |

Заместване на текст

Понякога е необходимо заместване на един текст с друг. За целта се използва функция SUBSTITUTE(текст,стар,нов), където текст е текстът, в който заместваме, стар – търсения подтекст, който ще заменим и нов – новия текст, който замества стария. Пример – ето как работи PROPER – не отчита, че апострофа не е нова дума.

| fx =PROPER(B4) | | | |
|-------------------|-------------------|--------------------|---|
| B | C | D | E |
| | | | |
| | | | |
| Company | Bad Proper Case | Better Proper Case | |
| STARBUCK'S COFFEE | Starbuck'S Coffee | Starbuck's Coffee | |
| MCDONALD'S | Mcdonald'S | Mcdonald's | |
| MICHAEL'S DELI | Michael'S Deli | Michael's Deli | |
| | | | |
| | | | |

За да го подобрим, използваме замяна на текст като заместваме апострофа с "qzx" и след използване на PROPER, връщаме обратно апострофа:

| 4 fx =SUBSTITUTE(PROPER(SUBSTITUTE(B4,"","qzx")),"qzx","") | | | | | | | |
|--|-------------------|--------------------|---|---|---|---|---|
| A | B | C | D | E | F | G | H |
| | | | | | | | |
| | | | | | | | |
| Company | Bad Proper Case | Better Proper Case | | | | | |
| STARBUCK'S COFFEE | Starbuck'S Coffee | Starbuck's Coffee | | | | | |
| MCDONALD'S | Mcdonald'S | Mcdonald's | | | | | |
| MICHAEL'S DELI | Michael'S Deli | Michael's Deli | | | | | |
| | | | | | | | |
| | | | | | | | |

Преброяване на специален символ в клетка

Целта е да се преброи колко пъти се среща един символ в текст. За да се направи, трябва:

1. Да се намери дължината на текста – използва се LEN()
2. Да се намери дължината на текста след премахване на всички срещания на символа – използва се LEN(SUBSTITUTE())
3. Изваждат се двете дължини и се добавя 1

Пример:

| fx =LEN(B4)-LEN(SUBSTITUTE(B4," ",""))+1 | | |
|--|---|----------------|
| A | B | C |
| | | |
| | | |
| | | Get Word Count |
| | The Quick Brown Fox Jumps Over The Lazy Dog. | 9 |
| | | |
| | | |

Добавяне на нов ред във формула

Когато създаваме графики в Excel е полезно да сложим нов ред във формула, за по-добра визуализация. Например:



Надписът по оста X има включен нов ред, за по-добър изглед. За целта е използвана функцията CHAR(10), която добавя нов ред между името и числото:

B3 fx =A3&CHAR(10)& C3

| | A | B | C | |
|---|-------|-----------|------------|------|
| 1 | | | | |
| 2 | | | Units Sold | |
| 3 | Mike | Mike 843 | 843 | |
| 4 | Mary | Mary 573 | 573 | 1200 |
| 5 | Ethan | Ethan 858 | 858 | 1000 |
| 6 | Emma | Emma 890 | 890 | 800 |
| 7 | Ava | Ava 602 | 602 | 600 |
| 8 | Mia | Mia 984 | 984 | |
| 9 | | | | |

Изчистване на странни символи в текст

Когато импортираме данни от външен източник като текстов файл или уеб, могат да се появят странни символи в данните. За изчистването им се ползва функцията CLEAN(текст). Тя връща изчистения текст. Често CLEAN() се използва съвместно с TRIM(), например:

| fx =TRIM(CLEAN(B3)) | | |
|-----------------------|---------------------------|----------------------|
| A | B | C |
| | | |
| | Store | Cleaned Text |
| | Detroit (Store #1)????? | Detroit (Store #1) |
| | Detroit (Store #2)????? | Detroit (Store #2) |
| | Detroit (Store #3)????? | Detroit (Store #3) |
| | Charlotte (Store #1)????? | Charlotte (Store #1) |
| | Charlotte (Store #2)????? | Charlotte (Store #2) |
| | Charlotte (Store #3)????? | Charlotte (Store #3) |

Добавяне на нули отзад на числата

Когато трябва да се подсигурим за точна дължина на числата, използваме добавяне на нули. Например, ако искаме Customer ID да бъде 10 цифри, можем да напишем:

| fx =LEFT(B4&"0000000000", 10) | | | |
|---------------------------------|--------------------|-----------------------------|---|
| A | B | C | D |
| | | | |
| | | | |
| | Customer ID | Pad to 10 characters | |
| | 5381656 | 5381656000 | |
| | 832 | 8320000000 | |
| | 23 | 2300000000 | |
| | 290 | 2900000000 | |
| | 2036 | 2036000000 | |
| | 5965 | 5965000000 | |
| | 6 | 6000000000 | |
| | 7457 | 7457000000 | |
| | 2903 | 2903000000 | |
| | 6137 | 6137000000 | |

В посочената по-горе формула създаваме текст, който обединява съдържанието на клетка и стринг(низ), състоящ се от 10 нули, след което използваме функция LEFT(), за да получим правилния резултат.

Форматиране на числа като текст

Често срещана обработка е слепване на текст с число. В този случай преобразуваме числото с функция TEXT(текст,формат), където текст е текстът, който трябва да се форматира, съгласно втория параметър формат. Пример:

| <div> <div> <div></div> <div></div> </div> <div> <div>✕</div> <div>✓</div> </div> <div> <div>fx</div> <div>=B3&": "&TEXT(C3, "\$0,000")</div> </div> </div> | | | | | |
|---|------------|----------------|---|------------------------|---|
| A | B | C | D | E | F |
| | | | | | |
| | Rep | Revenue | | Rep and Revenue | |
| | Gilbert | \$6 820 | | Gilbert: \$6,820 | |
| | Brown | \$5 205 | | Brown: \$5,205 | |
| | Tamburello | \$246 | | Tamburello: \$0,246 | |
| | Walters | \$7 136 | | Walters: \$7,136 | |
| | Alexander | \$2 921 | | Alexander: \$2,921 | |
| | Bradley | \$8 225 | | Bradley: \$8,225 | |
| | Dobney | \$5 630 | | Dobney: \$5,630 | |
| | Ellerbrock | \$7 994 | | Ellerbrock: \$7,994 | |
| | Hartwig | \$6 676 | | Hartwig: \$6,676 | |
| | Campbell | \$5 716 | | Campbell: \$5,716 | |
| | | | | | |
| | | | | | |
| | | | | | |

Друг начин за форматиране на числа е:

1. Десен бутон на мишката върху клетка и избиране на Format Cell
2. От Number Format таб се избира нужното форматиране
3. Избира се Custom от списъка с категории от лявата страна на Number Format диалог
4. Копира се синтаксиса от входното поле Type

Използване на функцията DOLLAR()

Ако числото, което ще се конкатенира с текст е парична сума, може да се използва функцията DOLLAR(текст,брой_знаци_след_запетаята). Обърнете внимание, че се работи с настройките на компютъра за парична единица. Например:

| <div> <div> <div></div> <div></div> </div> <div> <div>✕</div> <div>✓</div> </div> <div> <div>fx</div> <div>=B3&": "&DOLLAR(C3,0)</div> </div> </div> | | | | | |
|--|------------|----------------|---|------------------------|---|
| A | B | C | D | E | F |
| | | | | | |
| | Rep | Revenue | | Rep and Revenue | |
| | Gilbert | \$6 820 | | Gilbert: 6 820 BGN | |
| | Brown | \$5 205 | | Brown: 5 205 BGN | |
| | Tamburello | \$246 | | Tamburello: 246 BGN | |
| | Walters | \$7 136 | | Walters: 7 136 BGN | |
| | Alexander | \$2 921 | | Alexander: 2 921 BGN | |
| | Bradley | \$8 225 | | Bradley: 8 225 BGN | |
| | Dobney | \$5 630 | | Dobney: 5 630 BGN | |
| | Ellerbrock | \$7 994 | | Ellerbrock: 7 994 BGN | |
| | Hartwig | \$6 676 | | Hartwig: 6 676 BGN | |
| | Campbell | \$5 716 | | Campbell: 5 716 BGN | |
| | | | | | |
| | | | | | |
| | | | | | |